

# **.LYRIC & .LYRICX SPECIFICATION**

# **v1.0**

# .lyric spec:

Text-based lyrics for production with phrasing for AI singing.

Useful as a record for filing, as well as for use in AI music generation.

Includes:

- Header
  - optional features, such as: BPM, Key, etc
  - optional subsection for AI/Generation definitions
- Body
  - line breaks indicate phrasing

Supports “.txt” variants,  
such as “sample.lyric.txt”

1

1a

---  
Title: Elem-en Ellow Vo  
Artist: Kozzality

1b

[Generation]  
Styles: Edm, Goth, Industrial, Dubstep, Techno, Phonk  
Persona: Just Kozz  
---

2

[Verse]  
Epulote row span nisa con  
Ando go modo tray yo spawn  
Vosawn gane van cola mane  
Voss train man ola sane

[Chorus]  
Melon car stay voy  
Duran day  
Daylo cross ellow joy  
Dagra ba yay  
Elon shayow  
Colagris en ellow vo

[Verse 2]  
Enjelo base rico swan  
Carmelo daise micro vawn  
Torestry hawn ariso  
Comelise ray marino

[Chorus]  
Melon car stay voy  
Duran day  
Daylo cross ellow joy  
Dagra ba yay  
Elon shayow  
Colagris en ellow vo

[Bridge]  
Guess row heart  
Meya cart ay  
Terro part  
Genna karte  
...karte  
Enrin arte!

[Chorus]  
Melon car stay voy (melon karte)  
Duran day  
Daylo cross ellow joy (ellow joy)  
(Dagra ba yay)  
(Hey, hey)  
Colagris en ellow vo  
Colagris...  
en ellow vo!



Header section (metadata). Encapsulated by three ASCII hyphen-minus characters (-) on their own line (above and below.)

Required fields: Title, Artist

Possible Fields:

- **1a** (main section)

- Title
- Artist
- BPM
- Key
- TimeSignature
- Offset (offset to be applied to timing)
- Production (production notes)
- Version:  
document version. Acceptable values:
  - numeric semantic version (e.g., 1.0, 1.2)
  - descriptive label (Draft, Final, Radio Edit, etc)
- LXVersion: lyric spec version (1.0, 2.0, etc.)

- **1b** (AI/Generation subsection)

- Styles: styles prompt and generation instructions
- Persona: name of persona to be used
- Cover: title of song to be covered
- VocalGender: Male, Female, Neutral, Mixed
- Weirdness: 0.0 - 1.0
- StyleInfluence: 0.0 - 1.0
- Energy: 0.0 - 1.0
- TempoHint: descriptive
- Language: ISO Code, optional

## 2 -

Body: the lyrics. Possible sections

- Verse: Verse, Verse 2, etc.
- Pre-Chorus
- Chorus
- Bridge
- Intro
- Outro
- Final Chorus
- Spoken

# .lyricx spec:

Timed lyrics for performance, animation and AI singing timing.

Includes:

- Header
  - optional features, such as: BPM, Key, etc
  - optional subsection for AI/Generation definitions
- Body
  - numeric identifier
  - timestamp
    - optional: start and stop timestamp as MM:SS:mmm —> MM:SS:mmm (minutes, seconds, milliseconds)
  - lyric line (text)
  - lyric section (Verse, Chorus, etc)
  - line breaks within a timed entry indicate screen display only and do not imply phrasing.

Supports “.txt” variants, such as “sample.lyricx.txt”

1

1a

---  
Title: Elem-en ElLOW Vo  
Artist: Kozzality

1b

[Generation]  
Styles: Edm, Goth, Industrial, Dubstep, Techno, Phonk  
Persona: Just Kozz  
---

2

1  
00:48.633  
[Verse]  
Epulote row span nisa con

2  
00:51.667  
[Verse]  
Ando go modo tray yo spawn

3  
00:55.033  
[Verse]  
Vosawn gane van cola mane

4  
00:58.400  
[Verse]  
Voss train man ola sane

5  
01:09.333  
[Chorus]  
Melon car stay voy

6  
01:12.067  
[Chorus]  
Duran day

7  
01:13.167  
[Chorus]  
Daylo cross elLOW joy

8  
01:15.567  
[Chorus]  
Dagra ba yay

9  
01:17.333  
[Chorus]  
Elon shayow

# 1 -

Header section (metadata). Encapsulated by three ASCII hyphen-minus characters (-) on their own line (above and below.)

Required fields: Title, Artist

Possible Fields:

- **1a** (main section)
  - Title
  - Artist
  - BPM
  - Key
  - TimeSignature
  - Offset (offset to be applied to timing)
  - Production (production notes)
  - Version:
    - document version. Acceptable values:
      - numeric semantic version (e.g., 1.0, 1.2)
      - descriptive label (Draft, Final, Radio Edit, etc)
    - LXVersion: lyric spec version (1.0, 2.0, etc.)
- **1b** (AI/Generation subsection)
  - Styles: styles prompt and generation instructions
  - Persona: name of persona to be used
  - Cover: title of song to be covered
  - VocalGender: Male, Female, Neutral, Mixed
  - Weirdness: 0.0 - 1.0
  - StyleInfluence: 0.0 - 1.0
  - Energy: 0.0 - 1.0
  - TempoHint: descriptive
  - Language: ISO Code, optional

# 2 -

Body: Timed entries

- Numeric Identifier
- Timestamp
- Section (optional)
- Lyric line



# EBNF-1

## COMMON LEXICAL DEFINITIONS

```
newline      = "\n" ;
ws           = " " ;
digit        = "0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9" ;
alpha        = ? any unicode letter ? ;
alphanum     = alpha | digit ;

text_char    = ? any character except newline ? ;
text_line    = { text_char } ;

hyphen_line  = "-" , "-" , "-" ;

float_0_1     = "0"
               | "1"
               | "0." , digit , { digit }
               | "1.0" ;

integer      = digit , { digit } ;
```

# EBNF-2

## HEADER

```
header                = hyphen_line , newline ,
                        title_field ,
                        artist_field ,
                        { header_optional_field } ,
                        [ generation_block ] ,
                        hyphen_line , newline ;

header_optional_field = version_field
                        | bpm_field
                        | key_field
                        | timesignature_field
                        | offset_field
                        | production_field
                        | lxversion_field
                        | newline ;

title_field           = "Title:" , ws , text_line , newline ;
artist_field          = "Artist:" , ws , text_line , newline ;

bpm_field             = "BPM:" , ws , text_line , newline ;
key_field             = "Key:" , ws , text_line , newline ;
timesignature_field   = "TimeSignature:" , ws , text_line , newline ;

offset_field          = "Offset:" , ws , offset_value , newline ;
offset_value          = two_digits , ":" , two_digits , ":" , three_digits ;
two_digits            = digit , digit ;
three_digits          = digit , digit , digit ;
digit                 = "0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9" ;

production_field      = "Production:" , ws , newline , production_text ;
production_text       = { text_line_or_empty } ;
text_line_or_empty    = text_line | newline ;

version_field         = "Version:" , ws , text_line , newline ;
lxversion_field       = "LXVersion:" , ws , text_line , newline ;
```

# EBNF-3

## GENERATION BLOCK

```
generation_block =  
    "[Generation]" , newline ,  
    generation_fields ;  
  
generation_fields =  
    { generation_field } ;  
  
generation_field =  
    styles_field  
    | persona_field  
    | cover_field  
    | vocalgender_field  
    | weirdness_field  
    | styleinfluence_field  
    | energy_field  
    | tempohint_field  
    | language_field ;  
  
styles_field      = "Styles:" , ws , text_line , newline ;  
persona_field     = "Persona:" , ws , text_line , newline ;  
cover_field       = "Cover:" , ws , text_line , newline ;  
vocalgender_field = "VocalGender:" , ws ,  
                    ( "Male" | "Female" | "Neutral" | "Mixed" ) ,  
                    newline ;  
  
weirdness_field   = "Weirdness:" , ws , float_0_1 , newline ;  
styleinfluence_field = "StyleInfluence:" , ws , float_0_1 , newline ;  
energy_field      = "Energy:" , ws , float_0_1 , newline ;  
  
tempohint_field   = "TempoHint:" , ws , text_line , newline ;  
language_field    = "Language:" , ws , text_line , newline ;
```

## SECTION LABELS

```
section_name      = "Verse"  
                  | "Chorus"  
                  | "Pre-Chorus"  
                  | "Bridge"  
                  | "Intro"  
                  | "Outro"  
                  | "Final Chorus"  
                  | "Spoken" ;  
  
section_index     = ws , integer ;  
  
section_label     = "[" , section_name , [ section_index ] , "]" ;
```

# EBNF-4

## .LYRIC GRAMMAR

```
lyric_document =  
    header ,  
    lyric_body ;  
  
lyric_body =  
    { lyric_section } ;  
  
lyric_section =  
    section_label , newline ,  
    { lyric_line } ;  
  
lyric_line =  
    text_line , newline ;
```

## .LYRICX GRAMMAR

```
lyricx_document  = header ,  
                  lyricx_body ;  
  
lyricx_body      = timed_entry ,  
                  { separator , timed_entry } ;  
  
separator        = newline ;  
  
timed_entry      = integer , newline ,  
                  ( timestamp | time_range ) , newline ,  
                  [ section_label , newline ] ,  
                  timed_text ;  
  
timed_text       = text_line , newline ,  
                  { text_line , newline } ;
```

# EBNF-5

## TIMESTAMP (LYRICX)

```
timestamp      = minute , ":" , second , ":" , millisecond ;
minute         = digit , digit ;
second         = digit , digit ;
millisecond    = digit , digit , digit ;
time_range     = timestamp , ws , "--" , ">" , ws , timestamp ;
```

Canonical format:

MM:SS:mmm

# RULE SUMMARY-1

## HEADERS

1. Title and Artist are required
2. Title and Artist fields must appear first, in that order
3. Optional fields follow. Each label should be capitalized, by convention
  - a. BPM (recommended all-caps)
  - b. Key
  - c. TimeSignature (Pascalcase)
  - d. Offset (offset to be applied to timing)
  - e. Production (production notes)
  - f. Version
  - g. LXVersion ( capitalized exactly as shown)
  - h. By convention, "Version" and "LXVersion" should appear last in the optional section, and in that order
4. The "Generation" section comes last within the header block
5. "Generation" must be capitalized exactly as shown
6. "Generation" is optional and may be empty
7. Field order "Generation" is unrestricted
8. "Styles" within "Generation" is freeform and may include one or multiple empty lines
9. A single blank line must precede "Generation"

# RULE SUMMARY-2

## BODY

### .LYRIC

1. Sections are identified by labels using square brackets. The label should be capitalized. Example: [Verse]
2. Subsequent verses are numbered by order (e.g., [Verse], [Verse 2], [Verse 3])
3. Acceptable labels included:
  - a. Intro
  - b. Verse
  - c. Verse 2, Verse 3, etc
  - d. Pre-Chorus
  - e. Chorus
  - f. Bridge
  - g. Final Chorus
  - h. Outro
4. Section can be empty
5. By convention, it is recommended that there be an empty line between sections

# RULE SUMMARY-3

## BODY

### .LYRICX

1. Timed entries must follow this order:
  - a. Entry number.  
Entry numbers must:
    - i. Start at 1
    - ii. Increase sequentially by 1
    - iii. Be unique
    - iv. Appear in ascending order
  - b. Timestamp
  - c. Section label (optional)
  - d. Lyric line(s)
2. Lyric content may span multiple lines.
3. Lyric content must not contain empty lines, except at the end of an entry
4. Each entry must be seperated by exactly one empty line
5. Timestamp format must be MM:SS:mmm
6. An optional exit timestamp may be defined using the pattern:  
entry timestamp --> exit timestamp (two dashes followed by a greater-than symbol between timestamps)



# LISCENSING

Copyright 2026 Jade Rigby

Licensed under the Apache License, Version 2.0 (the "License");  
you may not use this file except in compliance with the License.  
You may obtain a copy of the License at:

<http://www.apache.org/licenses/LICENSE-2.0>